# 数値計算のスクエア
# シュミレーションと乱数
# （その 1）

Masato Shimura
jcd02773@nifty.com

2008 年 3 月 19 日

## 目次



## 1 指数分布と対数正規分布ー待ち行列

### 1.1 指数分布

シュミレーションでよく用いるランダム到着の確率事象は、(正規) 回数の側で見れば、ポアソン分布であり、時間感覚で見れば、指数分布になる。

分子の運動も指数分布となる。指数分布は無記憶性でありシュミレーションに活用される。

$f(x) = \lambda e^{-\lambda x}$

$E(x) = \dfrac{1}{\lambda}$

$V(x) = \dfrac{1}{\lambda^2}$

```
plot 10000 hist_count  1r59* exponentialrand 3000

   'stick' plot 10 hist_count exponentialrand 3000
```



図 1　指数分布

## 1.2　対数正規分布

$$lnN(\mu, \sigma^2, N) = \frac{1}{x\sigma\sqrt{2\pi}} exp^{\left(-\frac{1}{2}\frac{(x-\mu)^2}{\sigma^2}\right)}$$

対数を用いるので確率変数 $x$ の範囲は $(0<x<\infty)$ で負の値にはならない。

正規分布 $N(\mu, \sigma^2)$ を打ち出し、$s = \sigma t + \mu$ を求めて $e^s$ で作成する。



図 2　対数正規分布

```
  plot 0 hist_count 3000  rand_normal_ln 4.345 0.48
```

### 1.2.1 Script

```
 rand_normal_ln=: 4 : 0
NB. y is lnN( 4.345 0.48)
NB. y is N (rno 0 1 N)
'MU VAR '=: y
NML=: rno ; 0 1, x
1x1^ MU + (%: VAR)* NML
)

NB. grouping for histgram
hist_count=: 3 : 0
NB. usage:  plot (100) hist_count rno 4.3 0.48 10000
NB. x is pitch(times)// y is random number
; # L:0 (~: <. tmp)<;.1 tmp=. /:~ y
:
; # L:0 (~: <.&.(x &*) tmp)<;.1 tmp=. /:~ y
)

NB. ===========Norman thomson=======================
rnd=: (%~?)@(1e9&(#~)) NB. uniformed random number (0 1)
run=: +`(*rnd)/
rn=:-:-+/@run@:(0 1&,)NB. Normal distribution
NB. Usage: rno a b c//a(mean) b(standard deviation c(number)
NB. e.g.  nro 2 5 100 is N(2 5 100)
rno=: 3 : '({. y )+(1{y )*({: y )(rn &>@#)12'
```

## 1.3 到着時間とサービス時間のシュミレーション

issue:東北大学統計グループ

到着時間 サービス時間 対数サービス時間の計算を行う

| | 到着時間 | サービス時間 | 対数サービス時間 |
|---|---|---|---|
| 合計 | 2952 | 4889 | 217.448 |
| 平均 | 59.54 | 97.78 | 4.34897 |

| | | | | |
|---|---|---|---|---|
| 分散 | | 4019.41 | 5895.53 | 0.43661 |

**到着時間**

$$\frac{F(t+dt)-F(t)}{1-F(t)} = \frac{e^{\lambda t}-e^{\lambda(t+dt)}}{e^{-\lambda t}} = 1-e^{\lambda dt}$$

現時点から $dt$ 時間の間に次の客が到着する確率は、前の客の到着とは関係なく、常に $1-e^{\lambda dt}$ になる。

$$\frac{1}{\lambda} = 59.5$$

```
  1%59.5
0.016806
```

客の**到着時間**は、$\lambda = 0.0168$ の指数分布に従う。

**サービス時間**

対数分布で、確率変数 $Y = lnX$、平均 $\mu$、分散 $\sigma^2$ の正規分布に従うとき、サービス待ち時間の平均と分散は次のようになる。

| | | |
|---|---|---|
| 対数分 | $f(t) = \frac{1}{\sqrt{2\pi}\sigma x} e^{\{-\frac{(lnX-\mu)^2}{2\sigma^2}\}}$ | |
| 平均 | $e^{\mu+\frac{\sigma^2}{2}} (= e^{2\mu+\sigma^2})$ | 97.78 |
| 分散 | $e^{2\mu+2\sigma^2} - e^{2\mu+\sigma^2}$ | 5895.53 |

$$\left( \begin{array}{ll} e^{\mu+\frac{\sigma^2}{2}} & = 97.78 \\ e^{2\mu+2\sigma^2} - e^{2\mu+\sigma^2} & = 5895.53 \end{array} \right.$$

$2\mu + \sigma^2 = 2 \times ln97.78$

$a = 2 \times ln97.18 = 2\mu + \sigma^2$

$e^{2\mu+2\sigma^2} = 5895.53 + e^{2\mu+\sigma^2}$

$2\mu + 2\sigma^2 = ln(5895.53 + e^a) = ln(5895.53 + e^{2\times ln97.18})$

$$\left( \begin{array}{ll} 2\mu + \sigma^2 & = 9.16544 \\ 2\mu + 2\sigma^2 & = 9.64578 \end{array} \right.$$

この連立方程式を解くと、対数分布の平均と分散が求まる。生データがない場合は、この式により求める。

$a = 2 * ln97.78 = 2\mu + \sigma^2$

```
2* ^. 97.78
9.16544
```

$2\mu + 2\sigma^2 = ln(5895.53 + e^{2\times ln97.18})$

```
^. 5895.53+  1x1 ^ +: ^. 97.78
9.64578
```

連立方程式をクラメール法で解く

```
  dat=. 2 3 $ 2 1 9.16544 2 2 9.64578
   dat
2 1 9.16544
2 2 9.64578

   cr=: %.}:"1
   cr dat
          1 _1.11022e_15 4.34255
1.77636e_15            1 0.48034
```

客の到着時間は、$\lambda = \frac{1}{59.5} = 0.0168$ の指数分布に従う。
客のサービス時間は平均　4.34255　分散 0.48034 の対数分布に従う。

```
   machi_0 DATW
          1 _1.11022e_15  4.34255
1.77636e_15            1 0.480342

  a1=. 59.54;97.78;4019.41;5895.53
   machi_00 a1
          1 _1.11022e_15  4.34255
1.77636e_15            1 0.480342
```

先の指数分布と対数正規分布の乱数を指定個数打ち出し、到着時間とサービス時間のシュミレーションを行う。
乱数は $N.Thomson$ の Script を用いた。
$u$：(0, 1) の正規乱数とする
$t = \frac{1}{\lambda} ln(u)$ ： 指数分布
$N(m, \sigma^2)$ ：　正規分布 (へいきん：分散)
$s = \sigma t + m$
$e^s$　： 対数正規分布

```
a=. 59.54;97.78;4019.41;5895.53

  10 machi_sim0 a NB. unit is second
arrive(exp)arrive cumurative service(ln,real)
```

```
NB. ------------------------------------
   0.52249 38.6503 38.6503 4.17287 64.9016
  0.966477 2.03015 40.6805 4.66126   105.77
  0.491716 42.2647 82.9452 3.81293 45.2831
  0.738259  18.068 101.013 4.06519 58.2759
   0.24173 84.5429 185.556 4.86302 129.414
0.0464979 182.689 368.246 4.76546 117.385
   0.88253 7.44028 375.686 4.21626 67.7795
  0.807678 12.7173 388.403 5.31786 203.946
0.0833982 147.905 536.308 2.88564   17.915
  0.345582 63.2628 599.571 5.17127 176.139

x number of customers
```

y **平均（到着；サービス時間）；分散（到着；サービス時間）**

```
 10 machi_sim DATW NB.this type is also available
```

plot fulfill DATW




図 3　到着の状態

```
   machi_sim0=:4 : 0
if. 4=#y do.  NB. 4 parameter
'MN_AR MN_SV VR_AR VR_SV'=: y
'M0 M1'=:{:"1  machi_00 y  NB. parameter for ln normal random
else.         NB. native data
```

```
'M0 M1'=:{: "1  machi_0 y
end.
M11=: 1x1 ^S=:  M0 + (%: M1)* rno 0,1,x  NB. ln normal random
M12=: - MN_AR* ^. M10=. rnd x   NB. exponential random  NB. -1/lambda ln u
M10,.M12,.(; +/ L:0 <\ M12),.S,.M11
)
```

## 1.4　折り込み

### 1.4.1　到着時間とサービス時間のシュミレーション

生のデータからでも指数分布と対数正規分布のシュミレーションの 4 のパラメータからでも、次の様な到着時間とサービス時間の数列が作成できる。

```
 2 plot_machi1 DATW

(or) 2 plot_machi1 50 machi_sim 59.54;97.78;4019.41;5895.53
```

**x 台数 (ATM)、窓口数など**

**y 平均（到着；サービス時間）；分散（到着；サービス時間）**

```
   DATW
N0 arr wait NB.
 1  25  40
 2 170  37
 3 101 168
 4 308  58
 5  27  89
 6   2  52
```

2 plot_machi1 50 machi_sim0 a




図 4　待ち (青) と手すき（赤）の状態

### 1.4.2　折り込みアルゴリズム

折り込みアルゴリズムは次の方が簡易になる。

1. 待つときは並列よりも 1 列に並ばせる
2. 機械 (ATM) や窓口を中心に考える。
3. 最初に空く機械をソートで最初 (または最後）に持ってくる (FN)

手計算で一度丁寧に。

| | $M0$ | | | | | $M1$ | | | | |
|---|---|---|---|---|---|---|---|---|---|---|
| $NR$ | $lastopen$ | $arrive$ | $service$ | $nextopen$ | $fst.nxt$ | $lastopen$ | $arrive$ | $service$ | $nextopen$ | $fst.nxt$ |
| 0 | 65 | | | | $F$ | 232 | | | | $N$ |
| 1 | 65 | 296 | 168 | 464 | $F$ | 232 | 604 | 58 | 662 | $N$ |
| 2 | 464 | 631 | 89 | 720 | $N$ | 662 | 633($w$29) | 52 | 714 | $F$ |
| 3 | 720 | 744 | 59 | 803 | $N$ | 714 | 644($w$50) | 60 | 774 | $F$ |
| 4 | 803 | 789($w$14) | 42 | 845 | $F$ | 774 | 773($w$1) | 133 | 907 | $N$ |
| 5 | 845 | 810($w$35) | 113 | 958 | $N$ | 907 | 826($w$81) | 27 | 934 | $F$ |

```
  2 machi_anal DATW
nr. arv svce start end wait  NB. UNIT is sec.
NB. ---------------------
 0   25  40   25   65  _25
 1  195  37  195  232 _195
 2  296 168  296  464 _231
 3  604  58  604  662 _372
 4  631  89  631  720 _167
 5  633  52  662  714   29
 6  664  60  714  774   50
 7  744  59  744  803  _24
 8  773 133  774  907    1
 9  789  42  803  845   14
10  810 113  845  958   35
11  826  27  907  934   81
12  863  60  934  994   71
13  951 190  958 1148    7
14 1072  72 1072 1144  _78

NB. number/arrive time /service time/start service/end_service/waiting
NB. arrive start and end is cumurative(=real second)
NB. minus of wait is free time in operation
```

到着したときに機械や窓口をソートして早く空く方を予測しているので、実効率が多少上がっている。並列に並ぶときはソートを止めるアルゴリズムを用いる。

2 plot_machi2 DATW

3 plot_machi2 DATW

## 1.5 script

```
NB. ----------main------------------------
```




図 5　到着（青）と待ち (赤) 稼働（緑）の状態ー 2 台




図 6　到着（青）と待ち (赤) 稼働（緑）の状態ー 3 台

```
machi_anal=: 4 : 0
NB.  2 machi_anal DATW
NB. 3 machi_anal 50 machi_sim m0;m1;v0;v
NB.  59.54;97.78;4019.41;5895.53
DAT0=: 1 2 {"1 service0 y NB. del number index
MACHINE=. +/"1 TMP0=:  x {. DAT0 NB. next free time
ANS=. {(MACHINE) ,.(-{. "1 TMP0),.TMP0
for_ctr. i.  (# DAT0) - x do.
COMBI=.({. MACHINE),(ctr + x ){DAT0
'WAIT NEXTOPEN RENEW'=. calc_combi_sub COMBI
ANS=. ANS,<NEXTOPEN,WAIT, RENEW
MACHINE=. /:~ (}. MACHINE),NEXTOPEN
end.
```

```
0 1 2 5 3 4{"1  (i. # DAT0),.DAT0,.;("1),.  ANS
)

calc_combi_sub=: 3 : 0
'LAST_STAY NW_ARR NW_SVC'=. y
NB. last-open new-arrive new-servicetime
WAIT=. -/ LAST_STAY,NW_ARR NB. include minus
NXT_OPEN=. >./(LAST_STAY,NW_ARR) + NW_SVC NB. next open
WAIT;NXT_OPEN;(>./(LAST_STAY,NW_ARR)),NW_SVC
)
```

## 1.6 関数一覧

| file | machi_new.ijs | | | |
|---|---|---|---|---|
| catalogue | main | sub | explanation | usage |
| plot | plot_machi0<br>plot_machi1<br>plot_machi2 | | 到着の状態を表示<br>待ちと手空きの状態を表示<br>待ち, 処理状況、手空きの状態を表示 | 2 plot_machi0 y<br>2 plot_machi1 y<br>2 plot_machi2 y |
| analysis | machi_anal<br><br><br>fulfill<br>fulfill_service<br>fulfill_quere | calc_combi_sub<br><br><br>fulfill0<br>fulfill_quere<br>fulfill_quere0 | 待ち行列の処理のメイン<br><br><br>到着の状態の解析<br>処理状況<br>処理時の待ち状態 | x machi_anal y<br>x1 machi_anal x2 machi_sim y<br>fulfill y<br>x fulfill_service y<br>x fulfill_quere y |
| 確率 | machi_sim<br><br>machi_simx<br><br>machi_0<br><br>machi_00 | machi_sim0<br><br>machi_sim0x<br><br>machi_moment_sub<br><br>machi_moment_sub | 到着時間とサービス時間の乱数作成<br>（another）到着時間とサービス時間の乱数作成<br>実測テーブルから対数正規分布のパラメータを求める<br>平均と分散を与え対数正規分布のパラメータを求める | x machi_sim y<br><br>x machi_simx y<br><br>mach0_0 y<br><br>mach0_00 y |
| 乱数 | rud run rn rno | | 正規乱数 | N.Thomson による |
| sample file | touhoku_wait.csv | | ATM の到着とサービス時間 | DATW |

**■我国の待行列小史**　(森村英典「待ち行列事始め」より）

待ち行列の研究は電話が普及し始めた頃に始まったようである

1949　小島哲　通信呼理論の研究　科学振興社

が始まりで、最初の教科書となった。

1952 日科技連　OR 研究委員会　委員長　河田龍夫

1955　河田龍夫（東京工大）A probrem in the theory of Queres Rep.Stat.Appl.Res.JUSE,vol.3

1956 本間鶴千代(都立大、横浜市大)On the Server Quering Process withb a Particular Rep.Stat.Appl.Res.JUSE,vol.4

1957　OR 学会創設

1958　日本学術会議　数理科学総合研究班　第 6 班（研究代表　河田龍夫）で待ち行列とゲーム理論を研究。この待ち行列研究の 20 人の侍 (QR 会) の名簿に竹原清隆（日本道路公団　道路交通）が挙がっている。

## 2　J の組み込み乱数

classes/packages/stats/

random.ijs,statdist.ijs

Ver.6 で 乱数源にメルセンヌを採用し、強化されている。

New random number generators, including Mersenne Twister as the default RNG.

**■樅の木の植樹**　樅の木を一本植える。

分布のパラメータは次の 2 つがよく用いられる。

位置　 location parameter

木の高さと樹形 scale parameter

### 2.1　一覧

| | | | |
|---|---|---|---|
| 一様乱数 | rand01 v<br>rand11 | generate y random numbers in interval (0,1)<br>generate y random numbers in interval (_1,1) | |
| binom | binomialdist<br>binomialprob<br>binomialrand | y has 2 elements p,n: | |
| normal | normalprob<br>normalrand | y has 3 or 4 elements:<br>0 = mean of distribution<br>1 = standard deviation<br>2 = minimum result<br>3 = maximum result<br>(rand) with mean=0, standard deviation=1. | Box Muller method |
| Γ | gammarand | y has 2 elements p,n<br>0 = power parameter<br>1 = number of trials<br>if p=1 this is the exponential distribution | $f(x) = \frac{1}{\Gamma(\lambda)}\alpha^{\lambda}x^{\lambda-1}e^{-\alpha x}$ |

| | | | |
|---|---|---|---|
| $\beta$ | betarand | y has 3 elements p,q,n<br>p is probability of success in one trial<br>q is 1-p<br>n is number of trials | $f(x) = \dfrac{1}{B(p,q)} x^{p-1}(1-x)^{q-1}$<br>$E(x) = \dfrac{p}{p+q}$<br>$V(x) = \dfrac{pq}{(p+q)^2(p+q+1)}$<br>$B(p.q) = \dfrac{\Gamma(p)\Gamma(q)}{\Gamma(p+q)} = \dfrac{(p-1)!(q-1)!}{(p+q+1)!}$ |
| poisson | poissondist<br>poissonprob<br>poissonrand | y has 2 elements:<br>0 = mean of distribution<br>1 = maximum value to show<br>e.g. poissondist 2 10<br>= list of probabilities of values from 0 to 10<br>in poisson distribution of mean 2 | |
| exponential | exponentialrand | with mean=1. `F(x)=1-^-x`<br>y = number of trials | |
| cauchy | cauchyrand | $F(x) = 0.5 + (\arctan x)\%o.1$<br>y is number of trials | |

## 2.2 試運転

binomial random numbers

```
  binomialrand 0.4 20
1 1 1 1 1 0 1 1 1 0 0 1 1 1 0 1 1 0 0 0
```

```
binomialrand=: 3 : 0
'p n'=. y
r=. 2147483647
s=. <:p*r
s>?n#<:r
)
```

normal random numbers

Box Muller Method

互いに独立な一様乱数　$U_1, U_2$

$Z_1 = \sqrt{-2logU_1}cos(2\pi U_2)$

$Z_2 = \sqrt{-2logU_1}sin(2\pi U_2)$

```
normalrand=: 3 : 0
(2 o. +: o. rand01 y) * %: - +: ^. rand01 y
)
```

$Z_1$ is $cos$

without parameter

'stick' plot 100 hist_count normalrand 10000




図 7　normalrand

*poisson random numbers*

$f(x) = \dfrac{\lambda^x e^{-\lambda}}{x!}$

```
poissonrand=: 3 : 0
'm n'=. y
roll=. -@^.@rand01
r=. b=. m >: t=. roll n
i=. i.n
while. #i=. b#i do.
  b=. m >: t=. (b#t) + roll #i
  r=. (b + i{r) i } r
end.
r
)
```

'bar' plot 100 hist_count poissonrand 3 10000




図 8　*poisson*

*exponential random numbers*

$f(x) = \lambda e^{-\lambda x}$

```
exponentialrand=: 3 : 0
-^.rand01 y
)
```

$f(x) = e^{-\lambda x}$

'stick' plot 100 hist_count exponentialrand 10000




図 9　$e$

$\Gamma$ *random numbers*

$f(x) = \frac{1}{\beta^{\alpha}\Gamma(\alpha)} x^{\alpha} e(-\frac{x}{\beta})$

gammarand(p,n)

p is $\lambda(=\alpha)$, $\beta$ はなし）

p=1 is exponential rand /指数分布になる

独立な指数分布に従う確率変数の $n$ 個の和である。

$\alpha = 1$ の場合は平均 $\beta$ の指数分布 $(\lambda = \frac{1}{\beta})$

$\alpha$ が正の整数の場合はアーラン分布である。

$a = \frac{m}{2}, \beta = 2$ とすると自由度 $m$ の$\chi^2$ 分布

```
gammarand=: 3 : 0
'p n'=. y
r=. n#0
k=. p-i=. <.p
if. k do.
  r=. betarand k,(-.k),n
  r=. r * -^.rand01 n
end.
if. i do.
  r-^.*/rand01 i,n
end.
)
```

'stick' plot 10 hist_count gammarand 2 10000




図 10　$\Gamma(\lambda = 2)$

'stick' plot 10 hist_count gammarand 3 10000




図 11　$\Gamma(\lambda = 3)$

$\beta$ *randomnumbers*

$$f(x) = \frac{1}{B(a,b)} x^{a-1}(1-x)^{b-1}$$

$$where, B(a,b) = \int_0^1 t^{a-1}(1-t)^{b-1}dt$$

betarand(p1,p2,n)

p1 is $\lambda_1$, p2 is $\lambda_2$

```
betarand=: 3 : 0
'p q n'=. y
if. (1>p) *. 1>q do.
  b=. n#1
  r=. n#0
  whilst. 1 e. b do.
    m=. +/b
    x=. (rand01 m)^%p
    y=. x+(rand01 m)^%q
    t=. 1>:y
    z=. (t#x)%t#y
    i=. t#b#i.#b
    b=. 0 i } b
    r=. (z+i{r) i } r
  end.
else.
  s%(gammarand q,n)+s=. gammarand p,n
end.
)
```

'stick' plot 10 hist_count betarand 2 3 10000



図 12　$\beta(\lambda_1 = 2,\ \lambda_2 = 3)$

'stick' plot 10 hist_count betarand 4 6 10000



図 13　$\beta(\lambda_1 = 4,\ \lambda_2 = 6)$

cauchy random numbers

```
NB. random numbers in a cauchy distribution
NB. with F(x)=0.5+(arctan x)%o.1
NB.
NB. y  =  number of trials
cauchyrand=: 3 : 0
3 o. o. _0.5+rand01 y
)
```

'stick' plot 100 hist_count cauchyrand 10000




図 14　*cauchy*

```
 discreterand ((i.4);0.1 0.3 0.4 0.2);10
1 1 3 1 1 2 2 2 3 2
```

■Erlang　*Anger Krarup Erlang*(1878-1929) Denmark

コペンハーゲン大学出身、学校で数年間数学を中心に物理、天文、化学を研究し、教えた。数学協会でCTT(Copenhagen Telephone and Telegram) の主任技術者 *Jensen* から電話のコールと待ち時間の問題の解決を要請される。1908 年から CTT に在籍し、待ち行列の研究に従事。

1909　*The theory of probability and telephone conversation*

*Erlang* の待ち行列の理論は British Post Office を始めとし各国の電話会社で応用された。

対数などの数学関数の数表の作成にも業績を残した。

*issue* :(http://www-history.mcs.st-and.ac.uk/history/Biographies/Erlang.html/)

# 3　Reference

東北大学統計グループ「これだけは知っておこう！統計学」有斐閣　2002

# 付録 A

## A.1　離散一様分布

Discrete Uniform DIstribution
$P(X = k) = \dfrac{1}{k},\ \ (k = 1, 2, \cdots, n)$
n=6 ならばサイコロ

## A.2　2 項分布

$B(n, p)$
$P(X = k) = \begin{pmatrix} n \\ k \end{pmatrix} p^k (1 - p)^{n-k},\ \ (k = 1, 2, \cdots, n)$

# 付録 B　E.Show

plot n01pdf steps _5 5 100



図 15　normal pdf

plot n01cdf steps _5 5 100



図 16　normal cdf

### B.0.1　Γ

```
   gamma   1.5 5 6 7
0.886227 24 120 720


   6 ig0 0 5 10 20 30
0 46.0847 111.95 119.991 120
```

```
   6 incgam 0 5 10 20 30
0 0.384039 0.932914 0.999928 1
```

### B.0.2 Beta

```
   4 beta  2 3 4 1.5
0.05 0.0166667 0.00714286 0.101587

  4 4 2 4   beta  2 3 4 1.5
0.05 0.0166667 0.05 0.101587

   4 2 ib0 0 0.1 0.6 0.7 1
0 2.3e_5 0.016848 0.026411 0.05

   4 2 incbet 0 0.1 0.6 0.7 1
0 0.00046 0.33696 0.52822 1
```

### B.0.3 others

```
   5 0.2 bincdf i.6
0.32768 0.73728 0.94208 0.99328 0.99968 _.

  5 chisqcdf 0.831211 4.35146 11.0705 20.515
0.025 0.5 0.95 0.999

  3 10 fcdf 0.84508 3.70826 6.55231 12.5527
0.5 0.95 0.99 0.999

  2.3 poissoncdf i.5
0.100259 0.330854 0.596039 0.799347 0.916249
```